
Abstract

Streams, or infinite lists, have many applications in functional programming, and are naturally de-
fined using recursive equations. But how do we ensure that such equations make sense, i.e. that they
actually produce well-defined streams? In this article we present a new approach to this problem,
based upon the topological notion of contractive functions on streams. In particular, we give a sound
and complete representation theorem for contractive functions on streams, illustrate the use of this
theorem as a practical means to produce well-defined streams, and show how the efficiency of the
resulting definitions can be improved using another representation of contractive functions.

1 Introduction

Inductively defined types are a central concept in modern programming, with a wide
variety of applications. In recent years, it has become increasingly clear that the dual
notion of coinductive types are just as useful (Jacobset al., 2010). Examples of coinductive
types include infinite lists, infinite trees, transition systems, abstract datatypes, and more
generally, a wide variety of state-based dynamical systems (Jacobs & Rutten, 1997). In this
article we focus our attention on infinite lists, known more concisely asstreams.

In a functional language, streams are naturally defined using recursive equations. For ex-
ample, if we writex / xsfor the stream formed by prepending the valuex to the streamxs,
then the constant stream 1/ 1 / 1 / · · · can be defined byones= 1 / ones. But how do we
ensure that such equations make sense, i.e. that they actually produce well-defined streams?
For example, if we writetail for the function that removes the first value from a stream,
then the equationloop= tail loop is well-typed, but does not produce a well-defined stream
because unfolding the equation loops forever without producing any values.

Most approaches to this problem are based upon some form of guardedness, a syntactic
condition first proposed for use in process calculi by Milner (1989). In the case of stream
equations, the most basic form of guardedness states that all recursive uses of the stream
being defined must be guarded by the constructor/ , in the sense of occurring directly as
its second argument. For example, the equationones= 1 / onesis guarded, whereas the
equationloop= tail loop is not. While guardedness provides a simple syntactic means of
ensuring that stream equations are well-defined, it also excludes many valid definitions.

For example, if we writemergefor the function that selects alternative values from two
streams, then the stream of ones can also be defined byones= 1 / merge ones(tail ones),
but this equation fails to satisfy the guardedness condition.

In this article we present a new approach to the problem of ensuring that recursive stream
equations are well-defined, based upon the topological notion of contractive functions on
streams. More precisely, the article makes the following contributions:

• We give an accessible presentation and proof of Banach’s fixed point theorem for
contractive functions on streams, which provides a simple but powerful semantic
means of ensuring that stream equations are well-defined (section 4).
• We present a sound and complete representation theorem for contractive functions

on streams, which formalises such functions as precisely those whose output at any
point in time only depends on inputs at strictly earlier times (section 5).
• We illustrate the use of this theorem as a practical means to produce well-defined

streams (sections 6 and 7), and show how the efficiency of the resulting definitions
can be improved using another representation theorem (sections 8 and 9).

The article is aimed at readers who are familiar with the basics of functional program-
ming with streams, say to the level of chapter twelve of (Hutton, 2007), but no previous
experience with topological methods is assumed. The techniques are presented using a
Haskell-like syntax for a total functional language in which types are sets and programs
are total functions between sets, in the manner of (Turner, 1995).

2 Streams

In this section we review the idea of programming with streams, and introduce our notation.
We writeStream afor the type of streams of values of typea, which are constructed using
an infix operator/ that prepends a value to a stream of the same type, and are destructed
using functionsheadandtail that select and remove the first value from a stream:

head :: Stream a→ a
head(x / xs) = x

tail :: Stream a→ Stream a
tail (x / xs) = xs

Streams themselves are naturally defined using recursive equations. For example, if we
write Nat for the type of natural numbers then the constant stream 1/ 1 / 1 / · · · can be
defined as a single one followed by the stream itself:

ones :: Stream Nat
ones = 1 / ones

In turn, the stream of natural numbers can be defined by starting with zero, and then
mapping the successor function(+1) over each element of the stream itself:

nats :: Stream Nat
nats = 0 / map(+1) nats

map :: (a→ b)→ Stream a→ Stream b
map f (x / xs) = f x / map f xs

Unfortunately, not all recursive stream equations make sense as definitions of streams.
For example, the following equation is well-typed,

loop :: Stream a
loop = tail loop

but does not actually define a stream because unfolding the definition loops forever without
ever producing any values. Similarly, attempting to redefine

ones :: Stream Nat
ones = 1 / tail ones

is also invalid, because it produces a single one and then loops. On the other hand, not all
uses oftail are problematic. For example,onescan be defined as a single one followed by
the result of merging alternative elements from the stream itself and its tail:

ones :: Stream Nat
ones = 1 / merge ones(tail ones)

merge :: Stream a→ Stream a→ Stream a
merge(x / xs) ys = x / merge ys xs

However, if we swapped the order of the arguments tomergein the above definition for
ones, the definition again becomes invalid. This brings us to the central question of this
article: when does a recursive stream equation actually define a stream? In the next few
sections we introduce the technical machinery that underlies our approach.

3 Fixed points

In the previous section we informally reviewed the idea of streams and stream equations.
In this section we consider what these notions mean from a more formal perspective, in
terms of solutions of equations and fixed points of functions.

First of all, recall that inductive types are defined as theleastsolution of some equation.
For example, the typeNat of natural numbers can be defined as the least setX for which
there is a bijectionX ∼= 1 + X, where 1 is a singleton set with element∗, and + is
disjoint union of sets with injection functionsinl and inr. The right-to-left component
of the bijection,f :: 1+ Nat→ Nat, gives rise to the constructors for naturals by defining
Zero= f (inl ∗) and Succ n= f (inr n). In turn, the left-to-right componentg :: Nat→
1+Nat gives a form of case analysis, mappingZeroto inl ∗ andSucc nto inr n.

Dually, coinductive types are defined as thegreatestsolution of some equation. For
example, the equationX ∼= 1+ X also has a greatest solution, given by the typeNat∞

of natural numbers together with an infinite value defined byinf = Succ inf. In a similar
manner, the coinductive typeStream Aof streams of typeAcan be defined as the greatest set
X for which there is a bijectionX ∼= A × X, where× is Cartesian product of sets with pro-
jection functionsfst andsnd. The left-to-right component of the bijection,f :: Stream A→
A × Stream A, gives rise to the deconstructors for streams by defininghead xs= fst (f xs)
andtail xs= snd(f xs). In turn, the right-to-left componentg::A × Stream A→ Stream A
gives rise to the constructor for streams by definingx / xs= g (x,xs).

Just as types can be defined using equations, so too can values. Consider a recursive
equationxs= f xsthat defines a streamxsin terms of itself and some functionf . Any stream
that solves this equation forxsis called afixed pointof f . Hence, solving a stream equation
means finding a fixed point of a function on streams. However, not all such functions have
fixed points. For example, the functionmap(+1) has no fixed point, which corresponds to
the fact that the equationxs= map(+1) xsis not a valid definition for a stream. Moreover,
some functions have many fixed points. For example, the identity function has any stream
as a fixed point, which corresponds to the fact that the equationxs= xs is also an invalid
definition. Note that there is no general notion of ordering on streams, so it doesn’t make
sense to consider least or greatest fixed points in this context.

What then makes a valid definition? Our approach is to only consider functions on
streams that have auniquefixed point, denoted byfix f , which is adopted as the semantics
of the corresponding recursive equation. For example, the function(1 /) has a unique
fixed point given by the constant stream of ones, which corresponds to the fact that the
equationones= 1 / ones is a valid definition. In conclusion, the question of when a
recursive stream equation defines a stream can now be rephrased as follows: when does
a functionf :: Stream a→ Stream ahave a unique fixed pointfix f :: Stream a?

4 Contractive functions

Our approach to this question is based upon an idea from topology, in the form of contrac-
tive functions. In order to define this notion for streams, we first define a functiontakethat
returns the finite list comprising the firstn elements of a stream:

take :: Nat→ Stream a→ [a]
take0 xs = [ ]
take(n+1) (x / xs) = x : take n xs

For example,take3 ones= [1,1,1]. In turn, we will use the notationxs =n yswhen two
streams are equal for their firstn values. More formally, we define a family of equivalence
relations=n on streams of the same type as follows:

xs =n ys ⇔ take n xs= take n ys

For the purposes of proofs, however, we will usually find it more convenient to use the
following definition by explicit recursion:

xs =0 ys ⇔ True
(x / xs) =n+1 (y / ys) ⇔ x = y ∧ xs =n ys

The notion of contractivity can now be defined as follows:

Definition 1(contractive functions)
A function f :: Stream a→ Stream bis contractiveif xs =n ys implies f xs =n+1 f ys for
all natural numbersn and streamsxsandysof typea.

That is, a function on streams is contractive if whenever two streams are equal for their
first n elements, applying the function gives streams that are equal forn+ 1 elements. We
denote the type of contractive functions byStream a→c Stream b. The key property of
such functions is captured in the followed fixed point theorem (Banach, 1922):

Theorem 1(Banach’s theorem)
Every contractive functionf :: Stream a→c Stream ahas a unique fixed point.

In order to prove this result we first show that there can only be at most one fixed point
(uniqueness), and then that there is at least one fixed point (existence).

Proof (uniqueness). Suppose that a contractive functionf has two fixed pointsxsandys,
i.e. f xs= xs and f ys= ys. We are then required to show thatxs= ys. Using thetake
lemma(Bird & Wadler, 1988),xs= ys is equivalent to showing thatxs =n ys for all
natural numbersn, which property can then be verified by induction onn. The base case is
trivial, becausexs =0 ys is always true. For the inductive case, we calculate as follows:

xs =n+1 ys
⇔ { xsandysare fixed points off }

f xs =n+1 f ys
⇐ { f is contractive}

xs =n ys
⇐ { induction hypothesis}

True

Note that in the second step, contractivity is precisely the condition that is required to allow
the induction hypothesis to be used to complete the proof. ut

Proof (existence). We begin our construction by writinganya for an arbitrary stream of
typea, and defining a familySn of streams indexed by natural numbers:

S0 = anya

Sn+1 = f Sn

That is, the streamSn is given by then-fold application of the functionf to an arbitrary
stream; for example,S3 = f (f (f anya)). Note that existence of a streamanya requires that
the typea is non-empty, otherwise there are no streams of this type, but does not require
the axiom of choice because once we have a witness for the typea being non-empty this
value can simply be replicated to produce a constant stream.

The intuition behind the above definition is that the firstn elements of the streamSn will
form the firstn elements of a fixed point off . To formalise this idea, we first show that the
family Sn forms a chain-like structure, in the sense that the firstn elements of the stream
Sn coincide with the firstn elements of the next streamSn+1:

Lemma 1
Sn =n Sn+1.

Proof: by induction onn. The base case is trivial becausexs =0 ys is always true, while
the inductive case is once again a simple consequence of contractivity:

Sn+1 =n+1 Sn+2
⇔ { applyingS}

f Sn =n+1 f Sn+1
⇐ { f is contractive}

Sn =n Sn+1

⇐ { induction hypothesis}
True

ut

Now let us define an indexing operator !! that selects thenth element in a stream:

(!!) :: Stream a→ Nat→ a
(x / xs) !! 0 = x
(x / xs) !! (n+1) = xs!! n

Using this operator, together with the fact that a stream can be defined by specifying its
value for any index using the isomorphismStream a∼= Nat→ a (Altenkirch, 2001), we
then define a streams whosenth element is given by thenth element ofSn+1:

s!! n = Sn+1 !! n

This definition ensures that the firstn elements of the streamss andSn coincide:

Lemma 2
s =n Sn.

Proof: by induction onn. The base case is trivial, while for the inductive case we have:

s =n+1 Sn+1
⇐ { property of=n }

s =n Sn+1 ∧ s!! n = Sn+1 !! n
⇔ { definition ofs}

s =n Sn+1
⇐ { Lemma 1,=n is transitive}

s =n Sn

⇐ { induction hypothesis}
True

ut

The property of=n used above states if two streams are equal for their firstn elements
and their next elements are also equal, then they are equal forn+1 elements:

xs =n ys ∧ xs!! n = ys!! n ⇒ xs =n+1 ys

Proof: we assumexs= x / xs′ andys= y / ys′, and proceed by induction onn.

Base case:

xs =0 ys ∧ xs!! 0 = ys!! 0
⇔ { applying=0 and !!}

True ∧ x = y
⇔ { simplification}

x = y
⇔ { unapplying=1 }

xs =1 ys

Inductive case:

xs =n+1 ys ∧ xs!! (n+1) = ys!! (n+1)
⇔ { applying=n+1 and !!}

x = y ∧ xs′ =n ys′ ∧ xs′ !! n = ys′ !! n
⇐ { induction hypothesis}

x = y ∧ xs′ =n+1 ys′

⇔ { unapplying=n+2 }
xs =n+2 ys

ut
Using Lemma 2 it is now straightforward to show that the streams is a fixed point of the

functionf , thereby establishing the existence of a fixed point:

f s= s
⇔ { take lemma}
∀ n . f s =n s

⇔ { xs =0 ys is always true}
∀ n . f s =n+1 s

⇐ { Lemma 2,=n is transitive}
∀ n . f s =n+1 Sn+1

⇔ { applyingS}
∀ n . f s =n+1 f (Sn)

⇐ { f is a contractive}
∀ n . s =n Sn

⇔ { Lemma 2}
True

This completes the proof of Banach’s fixed point theorem for streams. ut
In summary, contractivity provides a sufficient condition for producing well-defined

streams, because contractive functions have precisely one fixed point. For example, a
simple calculation shows that the function(1 /) is contractive (that is,xs =n ys implies
1 / xs =n+1 1 / ys) and hence has a unique fixed point, which ensures thatones= fix (1 /)
is a valid definition for the constant stream of ones:

1 / xs =n+1 1 / ys
⇔ { applying=n+1 }

1 = 1 ∧ xs =n ys
⇔ { simplification}

xs =n ys

Similarly, the functionf = (0 /) ◦map(+1) is also contractive, which ensures thatnats=
fix f is a valid definition for the stream of natural numbers:

0 / map(+1) xs =n+1 0 / map(+1) ys
⇔ { applying=n+1 }

0 = 0 ∧ map(+1) xs =n map(+1) ys
⇔ { simplification}

map(+1) xs =n map(+1) ys
⇐ { property ofmap}

xs =n ys

The property ofmapused in the final step above states themap f preserves the equivalence
relation=n on streams, in the sense that:

xs =n ys ⇒ map f xs=n map f ys

Proof: by induction onn. The base case is trivial, while for the inductive case we have:

map f (x / xs) =n+1 map f (y / ys)
⇔ { applyingmap}

f x / map f xs=n+1 f y / map f ys
⇔ { applying=n+1 }

f x = f y ∧ map f xs=n map f ys
⇐ { extensionality, induction hypothesis}

x = y ∧ xs =n ys
⇔ { unapplying=n+1 }

x / xs =n+1 y / ys
ut

In contrast, the functiontail is not contractive. For example, in the case ofn = 0 the
contractivity property reduces tohead(tail xs) = head(tail ys), which is not true in general
because two arbitrary streamsxsandysof typea need not have the same second element,
except in the trivial cases where the typea is empty or contains a single possible value.
Hence, Banach’s fixed point theorem does not apply to the functiontail, and we reject the
equationloop= fix tail as being an invalid definition of a stream.

Contractivity ensures that functions have a unique fixed point, but raises some important
questions. First of all, does the converse theorem also hold: is every function on streams
with a unique fixed point contractive? The answer is no. For example, if we define

f :: Stream Nat→ Stream Nat
f xs = head(tail xs) / 1 / xs

then the functionf has a unique fixed point, given by the constant stream of ones, but is not
contractive. In particular, in the case ofn = 0 contractivity requires thathead(tail xs) =
head(tail ys), which as we have just seen is not always true.

And secondly, it is natural to ask what the notion of contractivity actually means, i.e.
what is being expressed in its definition? More generally, we can ask what kind of functions
are contractive, i.e. can the class of contractive functions be characterised in a precise
manner? The next section answers this question by providing a sound and complete repre-
sentation theorem for contractive functions on streams.

5 Representation theorem

Suppose that we view a stream as a value that may vary over time, where time is given
by a natural number index into the stream using the isomorphismStream a∼= Nat→ a.
Then contractive functions on streams are precisely those functions whose output value at
any time only depends on input values atstrictly earlier times. For example, the output at
time 3 can only depend on the inputs at times 0, 1 and 2, and no other times. In order to
formalise this idea, we introduce the following notion:

Definition 2(generating functions)
A generating functionis a function of type[a]→ b.

That is, a generating function maps a finite list of values to a single value. The intuition is
that the argument list records all earlier inputs for a contractive function (itsinput history),
and the result value is the next output of this function. We denote the type[a]→ b of
generating functions byGen a b. It is now straightforward to convert a generating function
into the corresponding contractive function on streams:

gen :: Gen a b→ (Stream a→c Stream b)
gen g(x / xs) = g [ ] / gen(g ◦ (x:)) xs

This definition expresses that the output value at any timen (counting from zero) is ob-
tained by applying the generating function to the firstn values of the input stream. For
example, applyinggen gto the stream 0/ 1 / 2 / · · · gives the output stream:

g [ ] / g [0] / g [0,1] / g [0,1,2] / · · ·

The validity of the definition ofgen, which requires that the resulting function on streams
is contractive, is established by the following result.

Lemma 3
If g is a generating function thengen gis contractive.

Proof: we are required to show thatxs =n ys ⇒ gen g xs=n+1 gen g ys, which we
verify by induction onn. In the base case, the precondition is trivially true, while the
postcondition is true becausehead(gen g xs) = g [ ] for any streamxs. For the inductive
case, the preconditionxs =n+1 ysimplies thatxs= x / xs′ andys= x / ys′ for somex, xs′

andys′ such thatxs′ =n ys′. We then verify the postcondition as follows:

gen g(x / xs′) =n+2 gen g(x / ys′)
⇔ { applyinggen}

g [ ] / gen(g ◦ (x:)) xs′ =n+2 g [ ] / gen(g ◦ (x:)) ys′

⇔ { applying=n+2 }
g [ ] = g [ ] ∧ gen(g ◦ (x:)) xs′ =n+1 gen(g ◦ (x:)) ys′

⇔ { simplification}
gen(g ◦ (x:)) xs′ =n+1 gen(g ◦ (x:)) ys′

⇐ { induction hypothesis,xs′ =n ys′ }
True

ut
Conversely, every contractive function can be represented as a generating function:

rep :: (Stream a→c Stream b)→Gen a b
rep f [ ] = head(f anya)
rep f (x : xs) = rep(tail ◦ f ◦ (x /)) xs

Given an input historyxsof lengthn, this definition expresses that the next output produced
by rep f will be the nth value in the result of applyingf to an argument stream that
begins withxs. Contractivity ensures that the result is independent of the remainder of
the argument stream, for which purposes we utilise an arbitrary stream in the base case.
For example, applyingrep f to the input history[0,1,2] gives the output value:

f (0 / 1 / 2 / any) !! 3

The validity of the recursive call in the definition ofrep, which requires that its argument
function is contractive, is established by the following result.

Lemma 4
If f is contractive thentail ◦ f ◦ (x /) is contractive.

Proof:

tail (f (x / xs)) =n+1 tail (f (x / ys))
⇐ { property oftail }

f (x / xs) =n+2 f (x / ys)
⇐ { f is contractive}

x / xs =n+1 x / ys
⇔ { applying=n+1 }

x = x ∧ xs =n ys
⇔ { simplification}

xs =n ys
ut

The property of the functiontail used above states that if two streams are equal for their
first n+1 elements then their tails are equal forn elements:

xs =n+1 ys ⇒ tail xs =n tail ys

Proof:

xs =n+1 ys
⇔ { applying=n+1 }

head xs= head ys∧ tail xs =n tail ys
⇒ { logic }

tail xs =n tail ys
ut

Using the two conversion functions, we can now formalise the idea that contractive and
generating functions are in one-to-one correspondence, i.e. every contractive function can
be uniquely represented by a generating function, and vice-versa.

Theorem 2(representation theorem)
The functionsgenandrep form an isomorphismGen a b∼= Stream a→c Stream b.

Proof: we verify the two parts of the isomorphism separately. For the first part,rep◦ gen=
id, we are required to show thatrep(gen g) xs= g xsfor any generating functiong and finite
list xs, which can be verified by induction onxs.

Base case:

rep(gen g) [ ]
= { applyingrep}

head(gen g anya)
= { applyinggenandhead}

g [ ]



ZU064-05-FPR paper 3 October 2011 15:0

Representing Contractive Functions on Streams 11

Inductive case:

rep(gen g) (x : xs)
= { applyingrep}

rep(tail ◦ gen g◦ (x /)) xs
= { applying◦ }

rep(λys→ tail (gen g(x / ys))) xs
= { applyinggenandtail }

rep(λys→ gen(g ◦ (x:)) ys) xs
= { eta reduction}

rep(gen(g ◦ (x:))) xs
= { induction hypothesis}

(g ◦ (x:)) xs
= { applying◦ }

g (x : xs)

For the second part,gen◦ rep= id, we are required to show thatgen(rep f) xs= f xsfor
any contractive functionf and streamxs, which can be verified by (guarded) coinduction
(Turner, 1995) onxs, which we assume has the formxs= x / xs′:

gen(rep f) (x / xs′)
= { applyinggen}

rep f [ ] / gen(rep f ◦ (x:)) xs′

= { applyingrep}
head(f anya) / gen(rep(tail ◦ f ◦ (x /))) xs′

= { coinduction hypothesis}
head(f anya) / (tail ◦ f ◦ (x /)) xs′

= { applying◦ }
head(f anya) / tail (f (x / xs′))

= { xs= x / xs′ }
head(f anya) / tail (f xs)

= { Lemma 5, f is contractive}
head(f xs) / tail (f xs)

= { streams}
f xs

ut
The lemma used in the second calculation above states that the first output value pro-

duced by a contractive function does not depend on the input stream.

Lemma 5
If f is contractive thenhead(f xs) = head(f ys) for any streamsxsandys.

Proof:

head(f xs) = head(f ys)
⇔ { unapplying=1 }

f xs =1 f ys
⇐ { f is contractive}

xs =0 ys
⇔ { applying=0 }

True
ut

6 Practical applications

Our representation theorem for contractive functions on streams has two main practical
applications. First of all, it simplifies the process of deciding if a function is contractive.
For example, it is easy to see that the function(1 /) is contractive, because the first output
(the value 1) requires no inputs, and each subsequent output only depends on the input at
one step earlier in time. That is,(1 /) acts as a one-step delay function. Conversely, the
function tail is not contractive, because the output at any time depends on the input at one
step later in time. That is,tail acts as a one-step lookahead function.

Secondly, our theorem provides a practical means of producing streams that are guar-
anteed to be well-defined. In particular, because contractive functions are in one-to-one
correspondence with generating functions, rather than defining a stream directly as the
fixed point of a contractive function (which requires verifying that the function is indeed
contractive) we can simply provide a generating function, from which the corresponding
function on streams (which is guaranteed to be contractive and hence have a unique fixed
point) can be generated automatically by applying the conversion functiongen from the
previous section. We encapsulate this idea by defining a new fixed point operator:

gfix :: Gen a a→ Stream a
gfix g = fix (gen g)

For example, the constant stream of ones can be defined usinggfix as follows (where the
function last returns the last element of a non-empty list):

ones :: Stream Nat
ones = gfix gones

gones :: Gen Nat Nat
gones[ ] = 1
gones xs = last xs

In this example, the generating functiongonesexpresses that the first value in the stream
(when the history list of previous values is empty) is one, and each subsequent value is the
same as the last, i.e. most recent, value in the history. Alternatively, we could simply state
that the next value in the stream is always the constant value one:

gones :: Gen Nat Nat
gones xs = 1

Note that the two definitions forgonesabove give rise to different contractive functions
under the conversion functiongen; it’s only their fixed points that are equal.

Similarly, the natural numbers can be defined using a generating function that expresses
that the first value is zero, and each subsequent value is the successor of the last:

nats :: Stream Nat
nats = gfix gnats

gnats :: Gen Nat Nat
gnats[ ] = 0
gnats xs = last xs+1

Alternatively, the next value is also given by the length of the history:

gnats :: Gen Nat Nat
gnats xs = length xs

Now consider the stream of Fibonacci numbers 0/ 1 / 1 / 2 / 3 / 5 / · · ·, in which the
first two values are zero and one, and each subsequent value is the sum of the preceding two
values. This behaviour translates directly into the following definition (where the function
penureturns the penultimate, i.e. next to last, element of a list):

fibs :: Stream Nat
fibs = gfix gfibs

gfibs :: Gen Nat Nat
gfibs[ ] = 0
gfibs[x] = 1
gfibs xs = penu xs+ last xs

As with the stream of natural numbers, the generating function forfibscan also be defined
in terms of the length of the history (where the indexing operator !! selects thenth element
of a list, in a similar manner to corresponding operator for streams):

gfibs :: Gen Nat Nat
gfibs xs = caselength xsof

0→ 0
1→ 1
n→ xs!! (n−2)+xs!! (n−1)

For our final example, let us return to the other definition for the constant stream of ones
in the introductory section, which defines the stream as a single one followed by the result
of merging alternative elements from the stream itself and its tail:

ones :: Stream Nat
ones = 1 / merge ones(tail ones)

Unlike the previous examples, it may not be immediately obvious how this behaviour
can be captured using a generating function. However, if we expand out the behaviour
of merge ones(tail ones) using indexing, we obtain the following stream:

ones!! 0 / ones!! 1 / ones!! 1 / ones!! 2 / ones!! 2 / · · ·

From this expansion it is now evident that the behaviour of themergecan be obtained
by indexing using aslowed downversion of the natural numbers that only increments

the number every other step (0,1,1,2,2, · · ·), which corresponds to the fact that the rate at
which values are consumed from each of the input streams tomergeis half the rate at which
values are produced to the output stream. Hence, the above definition foronestranslates
into using a generating function that indexes into the history at half its length (where the
division operator÷ rounds down to the nearest natural number):

ones :: Stream Nat
ones = gfix gones

gones :: Gen Nat Nat
gones[ ] = 1
gones xs = xs!! (length xs÷ 2)

In conclusion, many recursive stream equations (such as those forones, natsandfibs)
have a simple and natural definition in terms of generating functions, while some stream
equations (such as the alternative definition ofonesusingmerge) may require some thought
to express the desired behaviour in terms of the history of previous values of the stream.
However, defining streams in this manner has the important benefit that there is no longer
any need to check that the resulting stream is well-defined, because this is achievedby
constructionusing our representation theorem. Moreover, because our theorem establishes
an isomorphism between generating functions and contractive functions, we know thatany
stream that arises as the unique fixed point of a contractive function can be defined in this
manner, and hence that the technique is generally applicable.

7 Improving efficiency

We now turn our attention to the issue of efficiency. Using the Glasgow Haskell Compiler,
some simple experiments show that the standard definitions forones, natsandfibs using
recursive equations take linear time to produce their firstn values, as we would expect.
In contrast, our new definitions using generating functions take quadratic time, which is
unsatisfactory. There are two reasons for this behaviour. First of all, the functiongenthat
converts generating functions into contractive functions maintains the entire input history,
even though only a small part of this may be required to produce the next output value.
For example, in the case offibswe only require the last two values. And secondly, it takes
linear time to access the most recent values in the history list, due to the fact that it is
maintained in time order, i.e. with the most recent value at the end of the list.

The second problem can easily be solved by maintaining the input history in reverse
time order, i.e. with the most recent value at the start. In order to achieve this, we begin by
replacing the term(x:) that prefixes a value to the history in the definition of the conversion
functiongenby the term(++ [x]) that postfixes a value:

rgen :: Gen a b→ (Stream a→c Stream b)
rgen g(x / xs) = g [ ] / (g ◦ (++ [x])) xs

For example, applyingrgen gto input 0/ 1 / 2 / · · · gives the output:

g [ ] / g [0] / g [1,0] / g [2,1,0] / · · ·

In turn, we then eliminate the use of the inefficient append operator++ in the above
definition by using an extra argument to accumulate the history:

rgen :: Gen a b→ (Stream a→c Stream b)
rgen g = rgen′ g [ ]

rgen′ :: Gen a b→ [a]→ (Stream a→c Stream b)
rgen′ g ys(x / xs) = g ys/ rgen′ g (x : ys) xs

It is now straightforward to redefine generating functions to take their input history in
reverse order. For example, the generating function forfibs now has direct access to the
two most recent values in the input history by simple pattern matching:

rgfibs :: Gen Nat Nat
rgfibs[ ] = 0
rgfibs[x] = 1
rgfibs(x : y: zs) = y+x

Using this approach, the definitions for the streamsones, nats and fibs in terms of
generating functions now take linear time, with constant factors similar to the original
recursive definitions. It is natural at this point to wonder why we didn’t just use a reversed
history from the outset? The answer is that using the time ordered history gives a simpler
proof of our representation theorem for contractive functions.

While we have now regained the desired time performance, the space problem that
we raised at the start of this section still remains, namely that the entire input history
is not usually required to produce the next output value. However, it turns out that both
efficiency problems can be solved together in a simple manner using another representation
of contractive functions, this time in terms of infinite trees.

8 Generating trees

Let us writeTree a bfor the type ofa-branching infinite trees withb-labels in the nodes,
which are built using a prefix constructorNodethat forms a tree from a label of typeb and
a branching function of typea→ Tree a b, and are destructed using functionslabel and
branchesthat select the first and second components of a node:

label :: Tree a b→ b
label(Node y f) = y

branches :: Tree a b→ (a→ Tree a b)
branches(Node y f) = f

More formally, in set-theoretic terms the coinductive typeTree A Bcan be defined as the
greatest setX for which there is a bijectionX ∼= B × (A→ X). We will refer to values of
typeTree a basgenerating trees, because every such tree uniquely represents a generating
function. In order to formalise this idea, we first define a function that converts a generating

tree into the corresponding generating function:

gen′ :: Tree a b→Gen a b
gen′ (Node y f) [ ] = y
gen′ (Node y f) (x : xs) = gen′ (f x) xs

That is, the first output produced by the generating function (when its input history is
empty) is given by the labely of the generating tree, and subsequent outputs are given by
applying the branching functionf to the first valuex in the input history to obtain a new
tree that is used to process the remainder of the historyxs in the same manner.

Conversely, every generating function can be represented as a generating tree:

rep′ :: Gen a b→ Tree a b
rep′ g = Node(g [ ]) (λx→ rep′ (g ◦ (x:)))

This definition expresses that the label at each level in the resulting tree is given by applying
the generating functiong to the finite list of all previous branching values in the tree. For
example, the first few levels of the treerep′ g are as follows:

Node(g [ ]) (λx→
Node(g [x]) (λy→
Node(g [x,y]) (λz→
Node(g [x,y,z]) · · ·)))

Using the two conversion functions, we can now formalise that generating trees and
generating functions are in one-to-one correspondence, which is an instance of a general
result concerning the representation of functions with an inductive argument type — in this
case finite lists — using coinductive types (Altenkirch, 2001).

Theorem 3(representation theorem)
The functionsgen′ andrep′ form an isomorphismTree a b∼= Gen a b.

Proof: we verify the two parts of the isomorphism separately. For the first part,gen′ ◦
rep′ = id, we are required to show thatgen′ (rep′ g) xs= g xsfor any generating functiong
and finite listxs, which can be verified by induction onxs.

Base case:

gen′ (rep′ g) [ ]
= { applyingrep′ }

gen′ (Node(g [ ]) (λx→ rep′ (g ◦ (x:)))) [ ]
= { applyinggen′ }

g [ ]

Inductive case:

gen′ (rep′ g) (x : xs)
= { applyingrep′ }

gen′ (Node(g [ ]) (λx→ rep′ (g ◦ (x:)))) (x : xs)
= { applyinggen′ }

gen′ (rep′ (g ◦ (x:))) xs

= { induction hypothesis}
(g ◦ (x:)) xs

= { applying◦ }
g (x : xs)

For the second part,rep′ ◦ gen′ = id, we are required to show thatrep′ (gen′ t) = t for
any generating treet = Node f y, which can be verified by (guarded) coinduction ont:

rep′ (gen′ (Node y f))
= { applyinggen′ }

rep′ (λas→ caseasof {[ ]→ y; (x : xs)→ gen′ (f x) xs})
= { applyingrep′, simplification}

Node y(λx→ rep′ (gen′ (f x)))
= { coinduction hypothesis}

Node y(λx→ f x)
= { eta reduction}

Node y f
ut

Because the typeTree a bis coinductively defined, it comes equipped with a canonical
means of producing values of this type, in the form of anunfold operator (Meijeret al.,
1991; Gibbons & Jones, 1998). In order to define this operator, we first introduce the notion
of aco-algebra(Jacobs & Rutten, 1997) for generating trees:

type Coalg c a b = (c→ b, c→ a→ c)

That is, a co-algebra for the typeTree a bcomprises two functions that respectively turn a
value of typec into a value of typeb and a function of typea→ c. Using this notion, the
unfoldoperator for producing trees is then defined as follows:

unfold :: Coalg c a b→ c→ Tree a b
unfold(h, t) z = Node(h z) (λx→ unfold(h, t) (t z x))

That is, given a co-algebra(h :: c→ b, t :: c→ a→ c) and aseedvaluez:: c, the label of
the resulting tree is given by applyingh to the seedz, and the branching function is given
by applyingt to the seedz and the branching valuex :: a to obtain a new seed that is then
used to produce the remaining levels of the tree in the same manner.

9 Practical applications

Combining our two representation theorems with the use of theunfold operator provides
another means of producing contractive functions on streams. In particular, given a co-
algebra and a seed value, we first applyunfold to produce a generating tree, then apply
gen′ to convert this into a generating function, and finally applygento convert this into a
contractive function. We encapsulate this idea as follows:

generate :: Coalg c a b→ c→ (Stream a→c Stream b)
generate(h, t) z = gen(gen′ (unfold(h, t) z))

From the point of view of improving efficiency, however, it is desirable to fuse the three
functions in this definition together to give a direct recursive definitionforgenerate. In fact,
we can calculate the new definition directly from the above specification, by constructive
(guarded) coinductive on the argument streamx / xs:

generate(h, t) z(x / xs)
= { specification}

gen(gen′ (unfold(h, t) z)) (x / xs)
= { applyinggen}

gen′ (unfold(h, t) z) [ ] / gen(gen′ (unfold(h, t) z) ◦ (x:)) xs

We now simplify the head and tail of the resulting stream separately:

gen′ (unfold(h, t) z) [ ]
= { applyingunfold}

gen′ (Node(h z) (λx→ unfold(h, t) (t z x))) [ ]
= { applyinggen′ }

h z

and

gen(gen′ (unfold(h, t) z) ◦ (x:)) xs
= { applying◦ }

gen(λxs′→ gen′ (unfold(h, t) z) (x : xs′)) xs
= { applyingunfold}

gen(λxs′→ gen′ (Node(h z) (λx→ unfold(h, t) (t z x))) (x : xs′)) xs
= { applyinggen′ }

gen(λxs′→ gen′ (unfold(h, t) (t z x)) xs′) xs
= { eta reduction}

gen(gen′ (unfold(h, t) (t z x))) xs
= { coinduction hypothesis}

generate(h, t) (t z x) xs
ut

In conclusion, we have derived the following recursive definition:

generate :: Coalg c a b→ c→ (Stream a→c Stream b)
generate(h, t) z(x / xs) = h z / generate(h, t) (t z x) xs

It is useful now to think of the seed valuezas astatethat represents the input history of the
resulting contractive function. In this manner, the above definition expresses that the first
value in the output stream is given by applyingh to the current state (as it cannot depend on
the current or future input values to ensure contractivity), and the remaining output values
are given by applyingt to the current state and the first input valuex to obtain a new state
that is then used to process the tailxsof the input stream in the same way.

We encapsulate the idea of defining a stream as the unique fixed point of a contractive
function produced usinggenerateby defining a new fixed point operator:

cfix :: Coalg c a a→ c→ Stream a
cfix(h, t) z = fix (generate(h, t) z)

To define a stream usingcfix, we must first choose an appropriate state type to represent
the history of previous values in the stream, and then define a suitable starting value and
co-algebra for this type. Ideally, the state should be compact in terms of space, and the
co-algebra should be efficient in terms of time. For example, the natural numbers can be
defined using a state that comprises a single natural number that represents the next output
value, starting value zero, and a co-algebra(hnats, tnats) which expresses that the next
output and state are given by simply copying and incrementing the current state:

nats :: Stream Nat
nats = cfix(hnats, tnats) 0

hnats :: Nat→ Nat
hnats x = x

tnats :: Nat→ Nat→ Nat
tnats x = x+1

Similarly, the Fibonacci numbers can be defined using a state that comprises the next
two output values, starting value (0,1), and a simple co-algebra on this state:

fibs :: Stream Nat
fibs = cfix(hfibs, tfibs) (0,1)

hfibs :: (Nat,Nat)→ Nat
hfibs(x,y) = x

tfibs :: (Nat,Nat)→ Nat→ (Nat,Nat)
tfibs(x,y) = (y,x+y)

In terms of performance, the above definitions fornatsandfibsonce again take linear time,
but have the space advantage of using a small, finite state to produce the next output value,
rather than maintaining the entire output history for this purpose.

We conclude this section by showing howcfixcan also be used to define general purpose
functions to convert different representations of contractive functions into streams. First of
all, it is easy to convert a generating tree into a stream by exploiting the fact that two
projection functions on nodes naturally form a (terminal) co-algebra for such trees:

fromtree :: Tree a a→ Stream a
fromtree = cfix(label,branches)

It is also straightforward to convert a generating function into a stream, by means of a
co-algebra that applies the function to the empty history to produce the first output value,
and prepends this value to the current history to give a new generating function that is then
used to produce the remainder of the stream:

fromgen :: Gen a a→ Stream a
fromgen = cfix(hgen, tgen)

hgen :: Gen a b→ b
hgen g = g [ ]

tgen :: Gen a b→ a→Gen a b
tgen g x = g ◦ (x:)

Finally, we can produce a stream using a generating function that maintains its history
in reverse time order, by using a state that combines these two components:

fromrgen :: (Gen a a, [a])→ Stream a
fromrgen = cfix(hrgen, trgen)

hrgen :: (Gen a b, [a])→ b
hrgen(g,xs) = g xs

trgen :: (Gen a b, [a])→ a→ (Gen a b, [a])
trgen(g,xs) x = (g,x : xs)

10 Related work

In this section, we briefly survey a selection of other work that is most closely related to
our use of contractive functions as a means of ensuring that streams are well-defined.

Productivity.The notion of streams being defined in a productive manner first appears in
(Dijkstra, 1980). The first semantic treatment of productivity was given by Wadge (1981),
and later generalised by Sijtsma (1989). A number of techniques for ensuring productivity
have since been developed, including the use of syntactic guardedness (Coquand, 1994;
Telford & Turner, 1997), special type systems (Hugheset al., 1996; Sculthorpe & Nils-
son, 2009; Sculthorpe, 2011; McBride, 2011), custom algorithms (Endrulliset al., 2007;
Endrulliset al., 2008), and transformation techniques (Danielsson, 2010).

Unique fixed points.The idea of exploiting the fact that recursive stream equations may
have unique solutions dates back to early work on recursion operators (Hagino, 1987;
Malcolm, 1990), and has recently been promoted as a simple but powerful proof tech-
nique in its own right by Hinze (2008; 2010; 2011). In general, determining whether a
stream equation has a unique solution is undecidable (Endrulliset al., 2009), but a number
of partial algorithms have been developed, for example based upon the construction of
bisimulations (Capretta, 2010) and term rewriting systems (Zantema, 2010).

Contractive functions.The idea of defining the semantics of recursive stream equations
using contractive functions on metric spaces first appears in the work of de Bakker and
Kok (1985), and has since been applied to give semantics to many different kinds of lan-
guages and language features, including real-time systems (Müller & Scholz, 1997), PCF
(Escardo, 1999), guarded recursion (Birkedalet al., 2010a), mutable state (Birkedalet al.,
2010b), and reactive programs (Krishnaswami & Benton, 2011). A syntactic approach to
ensuring contractivity is developed in (Buchholz, 2005), and a type based approach in
(Krishnaswami & Benton, 2011). Generalisations of the underlying topological concepts
have also been considered (Matthews, 1999; Gianantonio & Miculan, 2003).

Representing functions on streams.Uustalu and Vene (2006) show how arbitrary and
causal functions on streams can be represented in terms of the zipper type for streams,
and explore the co-monadic nature of this representation. Hancock et al. (2009) show how
continuous functions on streams can be represented using infinite trees defined by mixed
induction/coinduction, and have generalised their result from streams to a large class of
coinductive types (Ghaniet al., 2009). However, none of these articles consider the stronger
notion of contractive functions on streams, which as we have seen is well-suited to the
problem of ensuring that recursive stream equations are well-defined.

11 Conclusion and further work

In this article we showed how Banach’s fixed point theorem for contractive functions
on streams can be formulated and proved correct using simple functional programming
techniques, presented representation theorems that formalise the temporal nature of con-
tractive functions and improve their performance, and showed how these theorems provide
a practical means of producing streams that are guaranteed to be well-defined.

There are many interesting topics for further work that can be considered, including
a range of generalisations (e.g. from streams to other terminal co-algebras, from SET to
other categories, from single to mutual recursion, from first-order to higher-order stream
functions, and from contractive functions to causal and continuous functions), the algebraic
structure of generating functions (e.g. how such functions can be composed and how this
can be done efficiently), the connection with theunfoldoperator for streams (e.g. our new
fixed point operators for producing streams can also be expressed usingunfold), more
precise typings for functions on streams (e.g. tracking how far they may look into the past,
or future, of their input stream), and the application of our techniques to implementing
stream programming languages such as FRP (Elliott & Hudak, 1997).

